EE324 Lab 6

Shubham Kar, 180070058

March 2021

1 Question 1

1.1 Part a

The steady state error is needed to be 0.489. We can obtain steady state value of $T(s)$ easily since it is a type 0 system. We just put $s = 0$ in $\left(\frac{KG}{1+KG}\right)$ and we obtain:

$$\frac{KG(0)}{1 + KG(0)} = (1 - 0.489) = 0.511 \tag{1}$$

$$\implies K = \left(\frac{1 - err}{err * G(0)}\right) \quad where\ err = 0.489 \tag{2}$$

$$\implies K = 150.478 \tag{3}$$

So, after obtaining this proportionality constant, we plot the step responses of the ideal step input and the step response of the system with the obtained proportionality constant shown in Figure 1.



Figure 1: Comparison of the step input and the step response

1.2 Part b

Since we have already got the damping ratio as 0.35, we can directly say that the closed loop dominant poles will have to lie at angles of $tan^{-1}\left(\frac{\sqrt{1-\zeta^2}}{\zeta}\right)$ from the x-axis. We can find the intersection of this line with the root locus to get the point at which we get the closed loop poles for which the damping ratio is 0.35. We get the value of K as **371.89** and the closed loop poles come out to be -14.878, $-2.061 + 5.516$ and $-2.061 - 5.516$. The closed loop poles obtained with the intersection of the root locus are shown in Figure 2.



Figure 2: Closed loop poles for getting damping ration as 0.35

1.3 Part c

The breakaway point comes out to be a point on the real axis as the root locus diverges as K increases from the open loop poles. The divergence happens between **-3** and **-4**. The point on the real axis is **-3.485**. The gain at that point comes out to be **2.1268**. The breakaway point of the root locus is shown in the Figure 3.

1.4 Part d

K was to vary in 0 to 1. Steps were taken to be 0.1 to generate 11 separate step responses. The closed loop poles as K increase from 0 to 1 are shown in Figure 4.

The step responses for each for variation in K are shown in Figure 5.

The steady state errors decrease as K increases. This is shown in Figure 6.



Figure 3: Breakaway Point on root locus



Figure 4: Closed loop poles on the root locus as K varies from 0 to 1

1.5 Part e

Here K changes from 1 to 1000 in steps of 100. The closed loop poles are plotted on the root locus and shown in Figure 7.
The step responses for each K is shown in Figure 8.
The settling times and the steady state errors as K increases from 1 to 1000 is shown in Figure 9.
Dips are seen in the settling time variation as the system changes from an overdamped one to an underdamped one as K increases from 1 to 100 and also we approximate third order system as a second order system owing more to these dips.



Figure 5: Step responses as K increases from 0 to 1



Figure 6: Steady state errors as K varies from 0 to 1

The system remains stable as K increases from 1 to 1000 as the closed loop poles do not cross the imaginary axis ever in this range of K.

The code for the lab is shown in the below listing:

```
clear();
s = poly(0, 's');

// Part a
G = 1/((s + 3)*(s + 4)*(s + 12));
sys = syslin('c', G);
err = 0.489;
K0 = ((1 - err)*(3*4*12))/err;
sys1 = (sys*K0)/.syslin('c', 1, 1);
t = 0:0.01:20;
y0 = csim('step', t, sys1);
scf();
clf();
plot(t, y0, 'b');
```



Figure 7: Closed loop poles on the root locus



Figure 8: Step responses comparison as K varies in 1 to 1000



(a) Settling times as K varies from 1 to 1000

(b) Steady state errors as K varies from 1 to 1000

Figure 9: Settling time and steady state error variation with Proportionality constant K

```
plot(t, ones(1, length(t)), 'r');
a = gca();
a.data_bounds = [0,0;t(length(t)), 1.1];
xgrid(2);
xlabel("Time t in sec");
ylabel("Step response");
title("Step input vs. step response");
xs2png(gcf(), 'Q1a.png');

// Part b
K = 0:0.01:400;
roots_den = zeros(length(K), 3);
i = 1;
for k = K
    sys1 = (sys*k)/.syslin('c', 1, 1);
    roots_den(i, :) = roots(sys1.den);
    i = i + 1;
end
zeta = 0.35;
phi = sqrt(1-zeta^2)/zeta;
temp = find(abs(imag(roots_den(:, 2))./real(roots_den(:, 2))) > phi);
mprintf(sprintf("Proportionality Constant = %0.2f", K(temp(1))));
mprintf("\n");
mprintf("Closed loop poles and T(s) for damping ration as 0.35 is:\n");
disp(roots_den(temp(1), :));
disp(sys*K(temp(1))/.syslin('c', 1, 1));
scf();
clf();
evans(sys, 600);
x = -6:0.1:0;
damp_line1 = -(phi).*x;
damp_line2 = phi.*x;
plot(x, damp_line1, 'r');
plot(x, damp_line2, 'r');
plot(real(roots_den(temp(1), :)), imag(roots_den(temp(1), :)), 'gx');
sgrid();
xs2png(gcf(), 'Q1b.png');

// Part c
[Kr, r] = krac2(sys);
mprintf("Gain at breakaway point is:")
disp(Kr);
mprintf("Breakaway point is at:")
disp(r);
scf();
clf();
evans(sys, 100);
sgrid();
plot(real(r), imag(r), 'rx');
xs2png(gcf(), 'Q1c.png');
// Part d
t = 0:0.01:20;
K1 = 0:0.1:1;
G0 = 1/(3*4*12);
y = zeros(length(t), length(K1));
ss_error = zeros(length(K1), 1);
```

```
i = 1;
scf();
clf();
evans(sys, 10);
sgrid();
for k = K1
    sys1 = (sys*k)/.syslin('c', 1, 1);
    y(:, i) = csim('step', t, sys1);
    roots_1 = roots(sys1.den);
    plot(real(roots_1), imag(roots_1), 'rx');
    ss_error(i) = (k*G0)/(1 + k*G0);
    i = i + 1;
end
ss_error = 1 - ss_error;
xs2png(gcf(), 'Q1d_1.png');
scf();
clf();
plot2d(t, y, 1:length(K1));
xlabel("Time t in sec");
ylabel("Step responses");
title("Comparison of step responses");
legend(['K=0.1', 'K=0.2', 'K=0.3', 'K=0.4', 'K=0.5', 'K=0.6', 'K=0.7', 'K=0.8', 'K=0.9', 'K=1']);
xgrid(2);
xs2png(gcf(), 'Q1d_2.png');
scf();
clf();
plot(ss_error);
xlabel("Proportionality constant(K)");
ylabel("Steady state error");
title("Steady state error vs. Proportionality constant");
a = gca();
a.x_ticks = tlist(["ticks", "locations", "labels"], (1:11), ['0', '0.1', '0.2', '0.3', '0.4', '0.5', '0.6', '0.7', '0.8', '0.9', '1']);
xgrid(2);
xs2png(gcf(), 'Q1d_3.png');
// Part e
K2 = [1, 100:100:1000];
G0 = 1/(3*4*12);
t = 0:0.01:20;
y = zeros(length(t), length(K2));
ss_error = zeros(length(K2), 1);
i = 1;
scf();
clf();
evans(sys, 1500);
sgrid();
for k = K2
    sys1 = (sys*k)/.syslin('c', 1, 1);
    y(:, i) = csim('step', t, sys1);
    roots_1 = roots(sys1.den);
    plot(real(roots_1), imag(roots_1), 'rx');
    ss_error(i) = (k*G0)/(1 + k*G0);
    i = i + 1;
end
ss_error = 1 - ss_error;
xs2png(gcf(), 'Q1e_1.png');
scf();
clf();
plot2d(t, y, 1:length(K2));
legend(['K=1','K=100', 'K=200', 'K=300', 'K=400', 'K=500', 'K=600', 'K=700', 'K=800', 'K=900', 'K=1000']);
xlabel("Time t in sec");
ylabel("Step responses");
title("Comparison of step responses for part e");
xgrid(2);
xs2png(gcf(), 'Q1e_2.png');
settling_time = zeros(length(K2), 1);
i = 1;
for k = K2
    temp = find(abs(y(:, i) - y(length(t), i)) > 0.05*y(length(t), i));
    disp(t(temp(length(temp))));
    settling_time(i) = t(temp(length(temp)));
    i = i + 1;
end
scf();
clf();
plot(settling_time);
xlabel("Proportionality constant(K)");
ylabel("Settling time in sec");
title("Settling time vs. Proportionality constant");
a = gca();
a.x_ticks = tlist(["ticks", "locations", "labels"], (1:11), ['1', '100', '200', '300', '400', '500', '600', '700', '800', '900', '1000']);
xgrid(2);
xs2png(gcf(), 'Q1e_3.png');
scf();
clf();
plot(ss_error);
xlabel("Proportionality constant(K)");
ylabel("Steady state error");
```

```
title("Steady state error vs. Proportionality constant");
a = gca();
a.x_ticks = tlist(["ticks", "locations", "labels"], (1:11), ['1', '100', '200', '300
    ', '400', '500', '600', '700', '800', '900', '1000']);
xgrid(2);
xs2png(gcf(), 'Q1e_4.png');
```